

Problem Statement

I changed the math in a roman font to be in a math italic font.

From [The Weekly Challenge - 155 Task #2: Pisano Period](#) retrieved on 2022-03-11 at 18:15-05:

Submitted by: [Mohammad S Anwar](#)

Write a script to find the period of the 3rd Pisano period.

In number theory, the n th Pisano period, written as $\pi(n)$, is the period with which the sequence of Fibonacci numbers taken modulo n repeats.

The Fibonacci numbers are the numbers in the integer sequence:

0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181, 6765, ...

For any integer n , the sequence of Fibonacci numbers $F(i)$ taken modulo n is periodic. The Pisano period, denoted $\pi(n)$, is the value of the period of this sequence. For example, the sequence of Fibonacci numbers modulo 3 begins:

0, 1, 1, 2, 0, 2, 2, 1,
0, 1, 1, 2, 0, 2, 2, 1,
0, 1, 1, 2, 0, 2, 2, 1, ...

This sequence has period 8, so $\pi(3) = 8$.

Raku Solution

```
# Use Raku version 6.d.
use v6.d;

# Find the period of the 3rd Pisano period.
my $modulo = 3;

# Define a Fibonacci series (called sequence in Raku).
# The last number in the series is the first number greater than 10,000.
my $fibonacci := 0, 1, -> $a, $b {$a + $b} ... -> $a {$a > 10_000};

# Compute the Fibonacci series modulo $modulo.
my @modulo = $fibonacci >>%>> $modulo;

# Find the period.
# Test periods from 2 to whatever.
for (2..*) -> $period
{
    # Divide @modulo into groups of period using ".rotor($period)"---groups
    # with fewer than period elements are ignored.
    # See
    # https://docs.raku.org/language/operators#infix\_eqv
    # for information about "eqv".
    # "[eqv] (1,2,3,4)" is equivalent to "1 eqv 2 eqv 3 eqv 4".
```

```
# "[eqv] ((1,2), (3,4))" is equivalent to "(1,2) eqv (3,4)".  
# If all groups are equal that is the period.  
[eqv] @modulo.rotor($period) and $period.say, last;  
}
```